



*Printed at the Mathematical Centre, 49, 2e Boerhaavestraat, Amsterdam.*

*The Mathematical Centre, founded the 11-th of February 1946, is a non-profit institution aiming at the promotion of pure mathematics and its applications. It is sponsored by the Netherlands Government through the Netherlands Organization for the Advancement of Pure Research (Z.W.O), by the Municipality of Amsterdam, by the University of Amsterdam, by the Free University at Amsterdam, and by industries.*

---

AMS (MOS) subject classification scheme (1970): 65D30

---

---

## Abstract

Numerical integration over a triangular domain is considered. Integration rules are given which are fully symmetric and nested in the sense that a higher degree formula contains all nodes used by a lower degree formula. An ALGOL 60 procedure is described which implements an adaptive algorithm based on 3-rd, 4-th and 5-th degree integration rules. Numerical experiences are reported.

## 1. Introduction

In this report we shall be concerned with cubature over a triangle; that is, for some given function  $f : \mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R}$ , we want to calculate the multiple integral

$$I = \iint_T f(x,y) \, dx \, dy \quad (1)$$

where the domain of integration  $T$  is a triangle, given by the coordinates of its vertices.

The method that will be described, also may be easily applied to integration over simplices in more dimensions. For the sake of simplicity, however, we will confine ourselves to the 2-dimensional simplex.

Following Cranley and Patterson [1969], we can distinguish between two basic approaches to numerical integration:

1. the use of "whole interval formulae".

A family of integration formulae, generally of high order, are applied to the whole domain of integration.

2. the use of "subdivision formulae".

The domain of integration is subdivided in a number of smaller domains, and an integration formula - generally of low order - is applied to each subdomain.

As pointed out by Cranley and Patterson for quadrature over a (one dimensional) interval, a more profitable approach may be found using families of whole interval formulae.

Recent experiences reported by Roothart and Fiolet [1972] show that in numerical quadrature very efficient and robust methods can be obtained by the use of a sequence of high order, nested quadrature formulae implemented in an adaptive algorithm.

We want to apply the same technique to integration over a triangular domain.

## 2. Location of the nodes

A sequence of integration formulae is called nested if the  $(i+1)$ st

member of the sequence contains all nodes of the  $i$ -th member. Using a nested sequence of integration formulae, no integrand evaluations are lost as one proceeds to the next member which often will be of a higher degree of approximation.

In section 5 we propose an adaptive algorithm based on a family of nested cubature formulae for the computation of the integral  $I$  (see eq.1).

In general, formulae for approximate numerical integration may be written in the form

$$\iint_T f(x,y) \, dx \, dy \approx \sum_{i=1}^q c_i f(x_i, y_i). \quad (2)$$

We only consider self contained formulae, i.e. all points  $(x_i, y_i)$  lie inside the domain of integration  $T$ . Moreover, we only want to consider formulae that are fully symmetric, i.e. the set of base points (nodes) is invariant under any affine transformation that maps the triangle  $T$  onto itself.

In order to describe these formulae, it is convenient to introduce the barycentric coordinates [cf. Mitchell 1972]

$$p_1(x,y) = \frac{l_{23}}{(l_{23})_1}, \quad p_2(x,y) = \frac{l_{31}}{(l_{31})_2}, \quad p_3(x,y) = \frac{l_{12}}{(l_{12})_3},$$

where e.g.  $l_{23}$  is the linear form of the straight line joining vertices 2 and 3 of the triangle (see Fig. 1);




Figure 1

$(l_{23})$  is the linear form evaluated at vertex 1 (Note:  $p_1 + p_2 + p_3 = 1$ ). In a fully symmetric formula, together with the node  $(p_1, p_2, p_3)$ , the nodes given by any permutation of the barycentric coordinates are also used.

Within the limitations of symmetry and self-containedness we want in addition that, after a division of the triangle  $T$  into four congruent

triangles  $T_1$ ,  $T_2$ ,  $T_3$  and  $T_4$ , each node for  $T$  is also a node for some sub-triangle  $T_i$ .




Figure 2.

This requirement is imposed in order to minimize the loss of integrand evaluations in the case where the adaptive algorithm has to subdivide the triangle.

By these observations we are led to the use of the following sets of nodal points.




Figure 3.

Many other sets may be constructed, but these are the simpler ones and they are sufficient for our purposes.

We will describe the location of the nodes in the triangle by means of the barycentric coordinates.

Figure 3a	coordinates	$(1/3, 1/3, 1/3)$	number	1.
3b		$(1, 0, 0)$		3.
3c		$(1/2, 1/2, 0)$		3.
3d		$(2/3, 1/6, 1/6)$		3.
3e		$(1/2, 1/4, 1/4)$		3.
3f		$(3/4, 1/4, 0)$		6.
3g		$(5/12, 5/12, 2/12)$		3.
3h		$(1/12, 1/12, 10/12)$		3.
3i		$(1/12, 4/12, 7/12)$		6.

Because of our requirements of full symmetry, equivalent nodes are obtained by permutation of the coordinates.

With these basic sets of nodes we can construct a hierarchical structure of compound sets of nodes (see figure 4).




Figure 4.



### 3. Integration formulae

Integration formulae corresponding to the sequence 'a' given in the scheme (figure 4) can be found in Silvester [1970]. Integration formulae corresponding to the node distributions given in the sequences b, bb, c and cc are given in the present report. Of course, the simpler formulae (1., 6. and 7.) are standard (cf. Abramowitz and Stegun [1965]) but the power lies in formulae of a higher degree.

Given a set of node points  $\{(x_i, y_i)\}$ , we have to calculate the constants  $\{c_i\}$  (see equation 2) such that we obtain optimal accuracy.

An integration rule is said to be of degree p if for any polynomial  $f(x,y)$  of degree less than or equal to p in x and y, the relation

$$\iint_T f(x,y) dx dy = \sum c_i f(x_i, y_i) \quad (3)$$

holds exactly. Hence, in order to obtain an approximation to I of degree p, we have to fulfill

$$\iint_T \sum_{0 \leq k+m \leq p} a_{km} x^k y^m dx dy = \sum_i c_i \sum_{0 \leq k+m \leq p} a_{km} x_i^k y_i^m \quad (4)$$

for all  $\{a_{km}\}$ , or equivalently

$$\iint_T x^k y^m dx dy = \sum_i c_i x_i^k y_i^m \quad \forall 0 \leq k+m \leq p. \quad (5)$$

It is easily seen that a polynomial  $P(x,y)$  of degree p in x and y is invariant under affine transformations, i.e. by any affine transformation

$$\begin{pmatrix} \xi \\ \eta \end{pmatrix} = \begin{pmatrix} \alpha \\ \beta \end{pmatrix} + \begin{pmatrix} b & c \\ d & e \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix}, \quad \det \begin{pmatrix} b & c \\ d & e \end{pmatrix} \neq 0$$

the polynomial  $P(\xi(x,y), \eta(x,y))$  will be of degree p if and only if the polynomial  $P(x,y)$  is. Hence, looking for integration formulae of degree p, we may transform our domain of integration T onto the triangle  $\{(0,0), (0,1), (1,0)\}$ . This enables us to formulate the requirements for

the coefficients  $\{c_i\}$ :

$$\iint_{\mathbb{T}} x^k y^m dx dy = \frac{k!m!}{(k+m+2)!} = \sum_{i=1}^q x_i^k y_i^m c_i \quad \forall 0 \leq k+m \leq p \quad (6)$$

These requirements constitute a system of  $(p+1)(p+2)/2$  linear equations with  $q$  unknown coefficients  $c_i$ .

Since we are concerned with fully symmetric integration rules it is clear that the weights  $c_i$  are the same for equivalent nodes  $(x_i, y_i)$ ; i.e. for any permutation of the barycentric coordinates  $x_i$ ,  $y_i$  and  $1-x_i-y_i$  we find

$$(x_i, y_i, 1-x_i-y_i) = \sigma(x_j, y_j, 1-x_j-y_j) \implies c_i = c_j,$$

where  $\sigma$  is the permutation operator.

This observation reduces system (6) to a linear system with  $r$  unknown coefficients,  $r$  being the number of independently chosen nodes in the triangle.

It will also be clear that all permutations of  $(x_i, y_i, 1-x_i-y_i)$  include the permutations of  $(x_i, y_i)$ . Hence condition (6) remains unchanged by permutations of  $k$  and  $m$ . This consideration reduces the order of the linear system that has to be solved.

Thus the system (6) is reduced to

$$\sum_{j=1}^r \left( \sum_{\text{perm. of}} x_i^k y_i^m \right) c_j = \frac{k!m!}{(k+m+2)!} \quad (7)$$

$$(x_i, y_i, 1-x_i-y_i) \quad \forall k, m \leq 0 \leq k \leq m \leq p$$

For every set of  $r$  freely chosen nodes in the triangle we may try to solve system (7) for some  $p$ .

For the sets of nodal points  $b$ ,  $bb$ ,  $c$ ,  $cc$  given in figure 4 we calculated the (unique) coefficients  $\{c_i\}$  for the largest possible degree  $p$ .

Table 1. Coefficients for sequence b, bb.

figure	p	$(x_i, y_i, 1-x_i-y_i)$	$c_i$	number
4.6	2	$(1/3, 1/3, 1/3)$	$3/4$	1
		$(1, 0, 0)$	$1/12$	3
4.7	3	$(1/3, 1/3, 1/3)$	$27/60$	1
		$(1, 0, 0)$	$3/60$	3
		$(1/2, 1/2, 0)$	$8/60$	3
4.8	4	$(1/3, 1/3, 1/3)$	$9/60$	1
		$(1, 0, 0)$	$1/60$	3
		$(1/2, 1/2, 0)$	$4/60$	3
		$(2/3, 1/6, 1/6)$	$12/60$	3
4.9	5	$(1/3, 1/3, 1/3)$	$2178/3780$	1
		$(1, 0, 0)$	$51/3780$	3
		$(1/2, 1/2, 0)$	$276/3780$	3
		$(2/3, 1/6, 1/6)$	$972/3780$	3
		$(1/2, 1/4, 1/4)$	$-768/3780$	3
4.10	5	$(1/3, 1/3, 1/3)$	$729/3780$	1
		$(1, 0, 0)$	$49/3780$	3
		$(1/2, 1/2, 0)$	$193/3780$	3
		$(2/3, 1/6, 1/6)$	$648/3780$	3
		$(3/4, 1/4, 0)$	$64/3780$	6

No integration rule of degree 6 can be obtained based on the set of 19 nodes given in triangle 11 of figure 4.

Table 2. Coefficients for sequence c, cc.

figure	p	$(x_i, y_i, 1-x_i-y_i)$	$c_i$	number
4.12	2	$(1/3, 1/3, 1/3)$	0	1
		$(2/3, 1/6, 1/6)$	$1/3$	3
4.13	3	$(1/3, 1/3, 1/3)$	$-18/45$	1
		$(2/3, 1/6, 1/6)$	$13/45$	3
		$(5/12, 5/12, 2/12)$	$8/45$	3
4.14	3	$(1/3, 1/3, 1/3)$	$-30/135$	1
		$(2/3, 1/6, 1/6)$	$63/135$	3
		$(1/12, 1/12, 10/12)$	$-8/135$	3
4.15	4	$(1/3, 1/3, 1/3)$	$-96/135$	1
		$(2/3, 1/6, 1/6)$	$-3/135$	3
		$(5/12, 5/12, 1/12)$	$66/135$	3
		$(1/12, 1/12, 10/12)$	$14/135$	3
4.16	5	$(1/3, 1/3, 1/3)$	$816/2835$	1
		$(2/3, 1/6, 1/6)$	$-63/2835$	3
		$(5/12, 5/12, 1/12)$	$-207/2835$	3
		$(1/12, 1/12, 10/12)$	$235/2835$	3
		$(1/12, 4/12, 7/12)$	$354/2835$	6

#### 4. The choice of a nested sequence of integration rules in an algorithm.

For computational purposes the sequences c, cc are less useful than sequences b, bb because of the negative coefficients which may cause some loss of accuracy.

In sequence b, bb we find two integration rules of degree 5: one rule with 16 nodal points, without negative coefficients, and one rule with 13 points and with one negative coefficient. In this case the negative coefficient can only cause a negligible loss of accuracy and by considerations of efficiency we are led to the use of the 13-point rule.

We notice that (after the computation of the barycenter of T) the computation of the coordinates of the nodal points of sequence 'b' turns out to consist of some additions and some divisions by 2 only.

5. A procedure for adaptive cubature over a triangle.

The procedure 'tricub' calculates the integral

$$\iint_T f(x,y) \, dx \, dy$$

with T a triangle in  $\mathbb{R}^2$ .

The procedure uses a nested sequence of four cubature formulae. These formulae which use 4, 7, 10 or 13 nodes, achieve approximations of degree 2, 3, 4 and 5 respectively. If the 5th degree approximation is not accurate enough, the triangle is subdivided into four congruent triangles, after which the algorithm is used recursively.

The procedure was written in ALGOL 60 and checked on an Electrologica X8 computer. The entries of the parameter list are as follows:  $(x_i, y_i)$ ,  $(x_j, y_j)$  and  $(x_k, y_k)$  are the vertices of the triangle T; f is a real procedure  $f(x,y)$  defining the integrand and 'ae' and 're' are the required absolute and relative error respectively.

It is advised to take for 'ae' and 're' values which are greater than the absolute and relative error in the computation of the integrand f.

```

real procedure tricub(xi,yi,xj,yj,xk,yk,g, re, ae);
value xi,yi,xj,yj,xk,yk, re, ae;
real xi,yi,xj,yj,xk,yk, re, ae; real proc g;
begin real surf, surfmin, xz, yz, xij, ylj, xjk, yjk, xki, yki, gi, gj, gk;

  real proc int(ax1, ay1, Af1, ax2, ay2, Af2, ax3, ay3, Af3,
                Bx1, By1, Bf1, Bx2, By2, Bf2, Bx3, By3, Bf3,
                px, py, Pf);
  value Bx1, By1, Bf1, Bx2, By2, Bf2, Bx3, By3, Bf3, px, py, Pf;
  real Bx1, By1, Bf1, Bx2, By2, Bf2, Bx3, By3, Bf3, px, py, Pf;
  begin real e, i3, i4, i5, A, B, C, bx1, by1, bx2, by2, bx3, by3,
           Cx1, Cy1, Cf1, Cx2, Cy2, Cf2, Cx3, Cy3, Cf3,
           Dx1, Dy1, Df1, Dx2, Dy2, Df2, Dx3, Dy3, Df3;

    A:= Af1 + Af2 + Af3; B:= Bf1 + Bf2 + Bf3;
    i3:= 3 × A + 27 × Pf + 8 × B;
    e:= abs(i3) × re + ae;

    if surf < surfmin ∨ abs(5 × A + 45 × Pf - i3) < e
    then int:= i3 × surf else
    begin Cx1:= ax1 + px; Cy1:= ay1 + py; Cf1:= g(Cx1, Cy1);
          Cx2:= ax2 + px; Cy2:= ay2 + py; Cf2:= g(Cx2, Cy2);
          Cx3:= ax3 + px; Cy3:= ay3 + py; Cf3:= g(Cx3, Cy3);
          C:= Cf1 + Cf2 + Cf3;
          i4:= A + 9 × Pf + 4 × B + 12 × C;

          if abs(i3 - i4) < e then int:= i4 × surf else
          begin bx1:= .5 × Bx1; by1:= .5 × By1;
                Dx1:= ax1 + bx1; Dy1:= ay1 + by1; Df1:= g(Dx1, Dy1);
                bx2:= .5 × Bx2; by2:= .5 × By2;
                Dx2:= ax2 + bx2; Dy2:= ay2 + by2; Df2:= g(Dx2, Dy2);
                bx3:= .5 × Bx3; by3:= .5 × By3;
                Dx3:= ax3 + bx3; Dy3:= ay3 + by3; Df3:= g(Dx3, Dy3);

                i5:= (51 × A + 2187 × Pf + 276 × B + 972 × C -
                      768 × (Df1 + Df2 + Df3))/63;

                if abs(i4 - i5) < e then int:= i5 × surf else
                begin surf:= .25 × surf;

                  int:=

                  int(bx1, by1, Bf1, bx2, by2, Bf2, bx3, by3, Bf3,
                      Dx1, Dy1, Df1, Dx2, Dy2, Df2, Dx3, Dy3, Df3,
                      px, py, Pf) +

```

```

int(ax1, ay1, Af1, bx3, by3, Bf3, bx2, by2, Bf2, Dx1, Dy1, Df1,
    ax1 + bx2, ay1 + by2, g(ax1 + bx2, ay1 + by2),
    ax1 + bx3, ay1 + by3, g(ax1 + bx3, ay1 + by3),
    .5 × Cx1, .5 × Cy1, Cf1) +
int(ax2, ay2, Af2, bx3, by3, Bf3, bx1, by1, Bf1, Dx2, Dy2, Df2,
    ax2 + bx1, ay2 + by1, g(ax2 + bx1, ay2 + by1),
    ax2 + bx3, ay2 + by3, g(ax2 + bx3, ay2 + by3),
    .5 × Cx2, .5 × Cy2, Cf2) +
int(ax3, ay3, Af3, bx1, by1, Bf1, bx2, by2, Bf2, Dx3, Dy3, Df3,
    ax3 + bx2, ay3 + by2, g(ax3 + bx2, ay3 + by2),
    ax3 + bx1, ay3 + by1, g(ax3 + bx1, ay3 + by1),
    .5 × Cx3, .5 × Cy3, Cf3);

surf:= 4 × surf

    end
  end
end int;

surf:= 0.5 × abs(xj × yk - xk × yj + xi × yj -
    xj × yi + xk × yi - xi × yk);
surfmin:= surf×re; re:= 30×re; ae:= 30×ae/surf;
xz:= (xi + xj + xk)/3; yz:= (yi + yj + yk)/3;
gi:= g(xi, yi); gj:= g(xj, yj); gk:= g(xk, yk);
xi:= xi×.5; yi:= yi×.5; xj:= xj×.5;
yj:= yj×.5; xk:= xk×.5; yk:= yk×.5;

tricub:= int(xi, yi, gi, xj, yj, gj, xk, yk, gk,
    xj+xk, yj+yk, g(xj+xk, yj+yk),
    xk+xi, yk+yi, g(xk+xi, yk+yi),
    xi+xj, yi+yj, g(xi+xj, yi+yj),
    .5 × xz, .5 × yz, g(xz, yz))/60

end tricub;

```

## 6. Numerical results

To demonstrate the behavior of the procedure tricub we report on a number of numerical experiments. The results obtained with 'tricub' are compared with those obtained with the one-dimensional integrators 'qad' and 'qadrat'. These procedures are described in Roothart and Fiolet [1972] 'qad' is an integrator of degree 5 (using 3th and 5th degree integration rules); 'qadrat' is an integrator of degree 16 (12th, 14th and 16th degree rules). In order to compute a double-integral with the aid of a one-dimensional integrator the integral is written as a repeated integral. This can be done in different ways and the numerical results depends strongly on this transformation.

The experiments reported here concern the double-integral over a right triangle and, therefore, two formulations as a repeated integral are obvious.

For each experiment we report:

- d : correct number of digits obtained  
i.e.  $-^{10}\log ((I_{\text{exact}} - I_{\text{computed}})/I_{\text{exact}})$ ;
- dr: number of digits required;
- fe: number of function evaluations.

All computations were performed on the Electrologica X8 computer of the Mathematical Centre.

### Problem 1

$$I = \iint_T \cos(x) \cos(y) \, dx \, dy; \quad T = \{(0,0), (0,\pi/2), (\pi/2,\pi/2)\}.$$

$$I^* = \int_0^{\pi/2} \int_x^{\pi/2} \cos(x) \cos(y) \, dy \, dx.$$

$$I^{**} = \int_0^{\pi/2} \int_0^y \cos(x) \cos(y) \, dx \, dy.$$



dr	tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
	d	fe	d	fe	d	fe	d	fe	d	fe
1.7	2.9	7	3.2	25	3.1	25	9.5	169	9.3	169
2.7	2.9	7	3.2	25	3.1	25				
3.7	4.7	10	5.1	57	5.1	53				
4.7	5.5	13	6.9	153	6.9	125				
5.7	5.5	13	7.7	357	8.7	409				
6.7	7.3	43	9.4	1117	9.6	1321				
7.7	7.8	133	9.8	2889	9.9	3305				
8.7	8.7	313	10.0	8129	10.0	8933	9.5	169	9.3	169

Problem 2

$$I = \iint_T f(x,y) \, dx \, dy; \quad T = \{(0,0), (0,-1), (-1/\sqrt{3}, -1)\}.$$

$$f(x,y) = \text{if } r > 1 \text{ then } 0 \text{ else } (1-r)^2(1+2r); \quad r = \sqrt{x^2+y^2}.$$

$$I^* = \int_{y=-1}^0 \int_{x=y/\sqrt{3}}^0 f(x,y) \, dx \, dy.$$

$$I^{**} = \int_{-1/\sqrt{3}}^0 \int_{y=-1}^{x\sqrt{3}} f(x,y) \, dy \, dx.$$

dr	tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
	d	fe	d	fe	d	fe	d	fe	d	fe
1	1.9	7	2.3	25	2.8	25	4.5	169	5.0	169
2	2.6	10	2.3	25	2.8	25	4.5	169	5.0	169
3	2.8	13	3.1	53	4.0	69	4.5	169	4.9	171
4	3.9	178	3.1	93	4.6	217	4.5	175	5.0	203
5	5.7	703	6.5	593	6.2	645	5.7	443	5.3	401
6	6.2	2299	7.1	1661	6.5	1797	7.0	745	7.4	1675
7	7.6	7357	7.1	3929	>10000		8.5	1395	8.2	2459

Problem 3

$$I = \iint_T f(x,y) \, dx \, dy; \quad T = \{(0,0), (0,-1), (-1/\sqrt{3}, -1)\}.$$

$$f(x,y) = \text{if } r > 1 \text{ then } 0 \text{ else } \exp\left(-\frac{1}{(1-r)^2}\right); \quad r = \sqrt{x^2+y^2}.$$

$$I^* = \int_{y=-1}^0 \int_{x=y/\sqrt{3}}^0 f(x,y) \, dx \, dy.$$

$$I^{**} = \int_{-1/\sqrt{3}}^0 \int_{y=-1}^{x\sqrt{3}} f(x,y) \, dy \, dx.$$

dr	tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
	d	fe	d	fe	d	fe	d	fe	d	fe
0.5	0.7	7	1.0	25	1.4	29	3.6	169	4.8	169
1.5	1.2	13	1.0	25	1.4	29	3.6	169	4.8	169
2.5	2.7	34	2.8	45	3.6	89	3.6	169	4.3	177
3.5	3.4	76	3.9	109	3.9	213	4.3	195	4.4	221
4.5	3.8	190	4.5	225	3.8	685	6.7	325	4.9	291
5.5	6.2	721	7.3	829	7.3	2637	6.7	325	7.1	561
6.5	7.3	1375	7.4	2257	8.6	6617	8.8	351	6.2	751
7.5	7.8	4072	7.8	6253	> 10000		8.3	481	7.7	1995

Problem 4

$$I = \iint_T f(x,y) \, dx \, dy; \quad T = \{(0,0), (0,-4/3), (-4/3\sqrt{3}, -4/3)\}.$$

$$f(x,y) = \text{if } r > 1 \text{ then } 0 \text{ else } (1-r)^n; \quad r = \sqrt{x^2+y^2}.$$

$$I^* = \int_{-4/3}^0 \int_{x=y/\sqrt{3}}^0 f(x,y) \, dx \, dy, \quad I^{**} = \int_{-4/3\sqrt{3}}^0 \int_{y=-4/3}^{x\sqrt{3}} f(x,y) \, dy \, dx.$$

n = 3

dr	tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
	d	fe	d	fe	d	fe	d	fe	d	fe
0	3.1	10	1.6	25	1.8	25	4.2	169	4.7	169
1	2.1	13	1.6	25	1.8	37	4.2	169	4.7	169
2	3.7	43	2.9	45	3.1	89	4.2	169	5.4	173
3	3.1	106	4.6	121	3.1	105	4.2	169	6.6	241
4	4.0	169	7.3	337	5.8	861	5.7	325	5.4	369
5	6.2	556	8.2	1125	7.1	2289	6.0	353	5.9	873
6	7.0	1186	7.8	3541	7.0	7141	7.6	559	7.6	2951
7	8.2	3145		>10000		>10000	9.0	801	8.1	6197
8	9.6	7576					9.7	1367	9.6	9581

n = 4

dr	tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
	d	fe	d	fe	d	fe	d	fe	d	fe
1	2.0	10	1.4	25	1.9	29	4.7	169	5.4	169
2	2.1	13	1.4	25	2.0	33	4.7	169	5.4	169
3	2.1	13	4.2	65	2.7	85	4.7	169	5.0	173
4	2.8	79	5.1	141	3.7	229	4.7	169	6.2	201
5	4.6	151	6.9	325	4.1	925	5.1	195	4.6	291
6	6.2	391	7.8	925	7.2	3361	6.4	325	6.4	1115
7	7.7	979	8.0	2721		>10000	6.8	351	7.3	1929
8	8.7	2227	9.8	8781			9.0	485	7.5	4371
9	9.3	5863		>10000			9.4	693	9.4	6749

5

tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
d	fe	d	fe	d	fe	d	fe	d	fe
1.2	10	0.9	25	1.7	29	5.3	169	5.2	169
1.7	13	0.9	25	1.7	37	5.3	169	5.2	169
2.5	37	3.5	65	3.4	153	5.3	169	4.8	173
3.4	109	3.5	113	3.3	229	5.3	169	6.0	199
4.4	256	6.7	397	3.3	449	5.3	169	4.3	251
6.0	469	6.8	869	7.0	3213	5.3	169	6.2	881
8.2	1543	8.7	2549	>10000		7.6	325	7.2	1677
8.2	4225	8.7	7485			9.2	351	7.4	3799
9.4	6976	>10000				9.7	511	9.3	9277

6

tricub: I		qad: I*		qad: I**		qadrat: I*		qadrat: I**	
d	fe	d	fe	d	fe	d	fe	d	fe
1.1	13	0.6	25	1.1	29	5.5	169	4.9	169
1.1	13	0.6	25	1.2	37	5.5	169	4.9	169
2.2	25	3.7	65	3.2	137	5.5	169	4.6	173
3.0	64	3.6	117	3.1	229	5.5	169	5.6	209
4.1	253	5.6	337	5.1	1389	5.5	169	4.1	251
6.2	508	6.7	929	5.0	3113	6.4	195	5.9	791
7.3	1516	7.4	2721	>10000		7.9	325	6.8	1431
8.4	4546	8.8	7485			8.9	351	8.6	3569
9.3	7807	>10000				9.6	481	9.1	6367

Conclusion:

The 5th degree two-dimensional integrator procedure 'tricub' compares favourably with the repeated use of a 5th degree one-dimensional integrator. The benefit is even greater since the calling of the two-dimensional integrator is simple and unique for each problem. When high precision is required, the use of the high-degree (16th degree) one-dimensional integrator is more favourable.

References

- Abramowitz M. and Stegun I. (ed.)  
Handbook of mathematical functions.  
Dover Publications, Inc. New York (1965).
- Cranley R. and Patterson T.N.L.  
On the automatic numerical evaluation of definite integrals.  
The Computer Journal 14 (1969) 189.
- Mitchell A.R.  
Variational principles and the finite element method.  
J. Inst. Maths. Applics., 9 (1972) 378.
- Roothart C.J. and Fiolet H.  
Quadrature procedures.  
Math. Centre Report MR 137/72.
- Silvester P.  
Symmetric quadrature formulae for simplexes.  
Math. Comp. 24 (1970) 95.
- Stroud A.H.  
Approximate calculation of multiple integrals.  
Prentice-Hall, Inc.. Englewood Cliffs, N.J. 1971.